NAG Fortran Library Routine Document

F07VGF (STBCON/DTBCON)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07VGF (STBCON/DTBCON) estimates the condition number of a real triangular band matrix.

2 Specification

```
 SUBROUTINE F07VGF(NORM, UPLO, DIAG, N, KD, AB, LDAB, RCOND, WORK, IWORK,
1                  INFO)
 ENTRY      stbcon (NORM, UPLO, DIAG, N, KD, AB, LDAB, RCOND, WORK, IWORK,
1                  INFO)
 INTEGER            N, KD, LDAB, IWORK(*), INFO
 real               AB(LDAB,*), RCOND, WORK(*)
 CHARACTER*1        NORM, UPLO, DIAG
```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine estimates the condition number of a real triangular band matrix A, in either the 1-norm or the infinity-norm:

$$\kappa_1(A) = \|A\|_1\|A^{-1}\|_1 \quad \text{or} \quad \kappa_\infty(A) = \|A\|_\infty\|A^{-1}\|_\infty.$$

Note that $\kappa_\infty(A) = \kappa_1(A^T)$.

Because the condition number is infinite if A is singular, the routine actually returns an estimate of the of the condition number.

The routine computes $\|A\|_1$ or $\|A\|_\infty$ exactly, and uses Higham's implementation of Hager's method (see Higham (1988)) to estimate $\|A^{-1}\|_1$ or $\|A^{-1}\|_\infty$.

4 References

Higham N J (1988) FORTRAN codes for estimating the one-norm of a real or complex matrix, with applications to condition estimation *ACM Trans. Math. Software* **14** 381–396

5 Parameters

1: NORM – CHARACTER*1 *Input*

On entry: indicates whether $\kappa_1(A)$ or $\kappa_\infty(A)$ is estimated as follows:

if NORM = '1' or 'O', $\kappa_1(A)$ is estimated;

if NORM = 'I', $\kappa_\infty(A)$ is estimated.

Constraint: NORM = '1', 'O' or 'I'.

2: UPLO – CHARACTER*1 *Input*

On entry: indicates whether A is upper or lower triangular as follows:

if UPLO = 'U', A is upper triangular;

if UPLO = 'L', A is lower triangular.

Constraint: UPLO = 'U' or 'L'.

3: DIAG – CHARACTER*1 *Input*

On entry: indicates whether A is a non-unit or unit triangular matrix as follows:

if DIAG = 'N', A is a non-unit triangular matrix;

if DIAG = 'U', A is a unit triangular matrix; the diagonal elements are not referenced and are assumed to be 1.

Constraint: DIAG = 'N' or 'U'.

4: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

5: KD – INTEGER *Input*

On entry: k, the number of super-diagonals of the matrix A if UPLO = 'U' or the number of sub-diagonals if UPLO = 'L'.

Constraint: KD ≥ 0.

6: AB(LDAB,*) – ***real*** array *Input*

Note: the second dimension of the array AB must be at least $\max(1, \mathrm{N})$.

On entry: the n by n triangular band matrix A, stored in rows 1 to $(k+1)$. More precisely, if UPLO = 'U', the elements of the upper triangle of A within the band must be stored with element a_{ij} in $\mathrm{AB}(k+1+i-j, j)$ for $\max(1, j-k) \leq i \leq j$; if UPLO = 'L', the elements of the lower triangle of A within the band must be stored with element a_{ij} in $\mathrm{AB}(1+i-j, j)$ for $j \leq i \leq \min(n, j+k)$. If DIAG = 'U', the diagonal elements are not referenced and are assumed to be 1.

7: LDAB – INTEGER *Input*

On entry: the first dimension of the array AB as declared in the (sub)program from which F07VGF (STBCON/DTBCON) is called.

Constraint: LDAB $\geq$ KD + 1.

8: RCOND – ***real*** *Output*

On exit: an estimate of the reciprocal of the condition number of A. RCOND is set to zero if exact singularity is detected or the estimate underflows. If RCOND is less than ***machine precision***, A is singular to working precision.

9: WORK(*) – ***real*** array *Workspace*

Note: the dimension of the array WORK must be at least $\max(1, 3 * \mathrm{N})$.

10: IWORK(*) – INTEGER array *Workspace*

Note: the dimension of the array IWORK must be at least $\max(1, \mathrm{N})$.

11: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed estimate RCOND is never less than the true value ρ, and in practice is nearly always less than 10ρ, although examples can be constructed where RCOND is much larger.

8 Further Comments

A call to this routine involves solving a number of systems of linear equations of the form $Ax = b$ or $A^T x = b$; the number is usually 4 or 5 and never more than 11. Each solution involves approximately $2nk$ floating-point operations (assuming $n \gg k$) but takes considerably longer than a call to F07VEF (STBTRS/DTBTRS) with one right-hand side, because extra care is taken to avoid overflow when A is approximately singular.

The complex analogue of this routine is F07VUF (CTBCON/ZTBCON).

9 Example

To estimate the condition number in the 1-norm of the matrix A, where

$$A = \begin{pmatrix} -4.16 & 0.00 & 0.00 & 0.00 \\ -2.25 & 4.78 & 0.00 & 0.00 \\ 0.00 & 5.86 & 6.32 & 0.00 \\ 0.00 & 0.00 & -4.82 & 0.16 \end{pmatrix}.$$

Here A is treated as a lower triangular band matrix with 1 sub-diagonal. The true condition number in the 1-norm is 69.62.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F07VGF Example Program Text
*     Mark 15 Release. NAG Copyright 1991.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, KDMAX, LDAB
      PARAMETER        (NMAX=8,KDMAX=NMAX,LDAB=KDMAX+1)
      CHARACTER        NORM, DIAG
      PARAMETER        (NORM='1',DIAG='N')
*     .. Local Scalars ..
      real             RCOND
      INTEGER          I, INFO, J, KD, N
      CHARACTER        UPLO
*     .. Local Arrays ..
      real             AB(LDAB,NMAX), WORK(3*NMAX)
      INTEGER          IWORK(NMAX)
*     .. External Functions ..
      real             X02AJF
      EXTERNAL         X02AJF
*     .. External Subroutines ..
      EXTERNAL         stbcon
*     .. Intrinsic Functions ..
      INTRINSIC        MAX, MIN
```

```
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F07VGF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, KD
      IF (N.LE.NMAX .AND. KD.LE.KDMAX) THEN
*
*        Read A from data file
*
         READ (NIN,*) UPLO
         IF (UPLO.EQ.'U') THEN
            DO 20 I = 1, N
               READ (NIN,*) (AB(KD+1+I-J,J),J=I,MIN(N,I+KD))
   20       CONTINUE
         ELSE IF (UPLO.EQ.'L') THEN
            DO 40 I = 1, N
               READ (NIN,*) (AB(1+I-J,J),J=MAX(1,I-KD),I)
   40       CONTINUE
         END IF
*
*        Estimate condition number
*
         CALL stbcon(NORM,UPLO,DIAG,N,KD,AB,LDAB,RCOND,WORK,IWORK,INFO)
*
         WRITE (NOUT,*)
         IF (RCOND.GE.X02AJF()) THEN
            WRITE (NOUT,99999) 'Estimate of condition number =',
     +        1.0e0/RCOND
         ELSE
            WRITE (NOUT,*) 'A is singular to working precision'
         END IF
      END IF
      STOP
*
99999 FORMAT (1X,A,1P,e10.2)
      END
```

9.2 Program Data

```
F07VGF Example Program Data
  4  1                        :Values of N and KD
  'L'                         :Value of UPLO
 -4.16
 -2.25   4.78
         5.86   6.32
               -4.82   0.16   :End of matrix A
```

9.3 Program Results

```
 F07VGF Example Program Results

 Estimate of condition number =  6.96E+01
```